

Automating Regression Output

Stata comes with a set of commands called **estimates**, which includes the following tools for manipulating regression or other estimation model results:

<code>estimates save file</code>	to save results to disk (as a .ster file)
<code>estimates use file</code>	to open results from disk
<code>estimates describe file</code>	to see details about results saved to disk
<code>estimates store name</code>	to save results to memory (temporary)
<code>estimates dir</code>	to see a list of stored estimates
<code>estimates title: text</code>	assigns a title to the current set of estimation results
<code>estimates notes: text</code>	assigns notes to the current set of estimation results
<code>estimates restore</code>	loads results into current/active estimation results
<code>estimates query</code>	tells you whether current estimates have been stored, and name
<code>estimates drop name</code> <code>estimates drop _all or</code> <code>estimates clear</code>	removes a set of estimates from memory removes all estimates from memory
<code>estimates notes list</code>	to see notes for estimates in memory
<code>estimates notes drop</code>	to drop notes for estimates in memory

estimates also includes the following for organizing results:

<code>estimates table names</code>	to display results in a table
<code>estimates stats names</code>	to display model selection statistics like the AIC and BIC (ML models)
<code>estimates for names</code>	to repeat postestimation command across models (e.g. test)

estimates table comes with lots of options for display formatting, including:

- `stats (scalarlist)` (anything contained in `e()`, for example)
- `star` for significance levels (can specify)
- `keep (coeflist)` or `drop (coeflist)`
- `b (%fmt)`, `se (%fmt)`, `t (%fmt)`, `p (%fmt)` - include coefficients, standard errors, t stats, p values (and specify format)

The **estout** package—which is user-created, and must be installed first—includes `estout`, `esttab`, `eststo` (an alternative to `estimates store`, and can be used as a prefix), `estadd`, `estpost` (can be used with commands like `summarize`, `tabulate`, or `correlate`).

Storing estimates using `eststo`:

- `_eststo:` or `_eststo [name]`: prefix before an estimation command will store results. Comes with options, including to add scalar statistics (`addscalars()`)--adds results to the `e()` scalars, such as postestimation results. The `[name]` is a shorthand name you can optionally assign to these estimation results.
- `eststo dir` -- displays a list of stored estimates (using `eststo`)
- `eststo drop` -- drops stored estimation results

Displaying and outputting results using `eststo`:

- `estout` -- to display results in a table
- `estout [what]` using *filename, options* -- to save formatted results to a file. The `[what]` typically includes a list of stored estimates, or “`_all`”

Options for `estout` are numerous, and include:

- `cells()`: contents of table cells. Such as `b`, `se`, `var`, `t`, `z`, `p`, `ci`, `ci_l`, `ci_u`
 - `star`: attach significance stars
 - `fmt`: format of each element in cells
- `drop` -- drop specific coefficients
- `keep` -- keep specific coefficients
- `order` -- order of coefficients in table
- `indicate` -- yes/no indicators for presence of parameters
- `stats` -- display summary statistics at the bottom of the table
 - `fmt`: format for summary statistics
- `starlevels` -- set thresholds for significance
- `label` -- use variable labels
- `title` -- add title to the table
- `replace` -- replace existing file
- `append` -- append existing file
- `style` -- can be `smcl`, `tab` (tab delimited), `fixed`, `tex`, `html`

Display and outputting results using `esttab`:

- `esttab [namelist]` using *filename, options*: similar to `estout`, but document format can be `smcl`, `fixed`, `tab`, `csv`, `scsv`, `rtf`, `html`, `tex`, `booktabs`

Related commands include **outreg2** and `putexcel` (new in Stata 13). **outreg2** is even more powerful than `estout`. The syntax for outputting results using `outreg2` is as follows:

- `outreg2 [estlist]` using *filename* [, options]

Like `estout`, the options for `outreg2` are numerous, and include:

- `stats()`: statistics reported for each variable. These include `coef`, `se`, `var`, `tstat`, `pval`, `ci`, `ci_low`, `ci_high`, and many others
 - `asterisk()`: attach significance stars, e.g. `asterisk(stats(coef))`
 - `dec(#)` or `fmt(type)`: format of elements, such as decimal places
 - `noparen` – specifies no parentheses around standard errors
- `drop` -- drop specific coefficients
- `keep` -- keep specific coefficients
- `label()` -- use variable labels
- `title()` -- add title to the table
- `replace` -- replace existing file
- `sideway` -- place standard errors, t-statistics, etc. next to coefficients rather than below

One of the advantages to `outreg2` is the types of files it can save to (specified in the options): `dta` (a Stata file), `word`, `excel`, `tex`, `text`, `comma`

```

sysuse auto.dta
reg price mpg
estimates store model1

reg price mpg headroom trunk
estimates store model2

reg price mpg headroom trunk foreign
estimates store model3

estimates dir

estimates table _all

estimates table _all, b(%6.1f) se(%6.1f)

estimates table _all, b(%6.1f) star title(Auto price
regressions)

estimates drop _all

** now use eststo set of commands

_eststo modelA: reg price mpg
_eststo modelB: reg price mpg headroom trunk
_eststo modelC: reg price mpg headroom trunk foreign

eststo dir

estout

estout, label

estout using Results.tab, label

estout using Results.tab, label cells(b se) replace

estout using Results.tab, label cells(b(star fmt(%6.1f))
se(par("`="("`'`")"'') fmt(%6.1f))) replace

** esttab

esttab using Results2.rtf, label cells(b(star fmt(%6.1f)) se(par
fmt(%6.1f))) replace

```