

Useful Stata Commands for Simulation

1. Loops – loops allow you to repeat a section of code in your do file, typically changing one or more elements of that code with each iteration. The part of the code that changes from one iteration to the next is a “local macro variable.” You can loop over a sequence of values, or a list of strings or variable names. The appropriate syntax is below where *j* is an arbitrary name for the local macro (you can choose this). It stands in for the value on each iteration:

<code>foreach <i>j</i> in ...</code>	when looping over a list of strings/values
<code>foreach <i>j</i> of varlist ...</code>	when looping over a list of variables
<code>forvalues <i>j</i>=#/#</code>	when looping over a sequence of values

Loops require an open bracket at the end of the first line and a closing bracket on a line by itself at the conclusion of the loop. For example:

```
forvalues j=1/10 {  
    display `j'  
}  
foreach color in red yellow green {  
    display "`color'"  
}  
foreach var of varlist income age {  
    sum `var'  
}
```

2. Local macro variables – in the examples above, *j*, *color*, and *var* are all local macros. After a local macro is defined, every reference to it must be surrounded with single quotes. Moreover, the opening quote must be “sloped down and to the right” (grave accent); the closing quote must be “sloped down at to the left” (acute accent). In the second example above, the local macro *color* is resolved using single quotes; since the resulting value of *color* is a string (the words red, yellow, and green) and we want to display them, the value must be surrounded by double quotes, as with any string. In the first example, *j* is numeric, so no double quotes are needed when it is used in the display command.

Local macro variables can also be defined outside of loops. The key difference between local and global macros (see next item) is that local macro variables are only “active” while your do-file is running. When the do-file stops, they are no longer in memory. The example below creates two local macro variables and then uses them in a display command.

```
local j=10  
display `j'+5  
local name Vanderbilt  
display "`name' is a top university"
```

3. Global macro variables – global macros work just like local macros, the main difference being that they remain in memory outside of your do-file. (They are retained until you erase

them, replace them, or exit Stata). Unlike local macros, after a global macro is defined, every reference to it must be preceded by a dollar sign \$. Some basic examples follow:

```
global myname "Sean Corcoran"
display "My name is $myname"
global workdir "C:\My Documents\Stats Class"
cd "$workdir"
```

Again, if your global macro resolves as a string value and the string value would normally require the use of double quotes, then be sure to include double quotes as above.

4. Macro utilities – there are lots of commands that allow you to manage macros, but several useful ones are:

macro dir	list all macros currently in memory
macro drop <i>myname</i>	drop the macro called <i>myname</i>
macro drop _all	drop all user-defined macros

Stata also has system-created macros that contain things like the current data's filename, or variables used in previous commands. You will see these when your data is memory and you type macro dir.

You can even use “extended macros” which can grab other features of your data. For example, the command below creates a local macro called *inclabel* that will contain the variable label for the variable called *income*. Type `help extended_fcn` for a list of possible extended macros.

```
local inclabel : variable label income
```

5. Preserve and restore – Stata allows you to **preserve** your data in its current state, make changes to it, and then **restore** the data back to the point at which you preserved it. For example:

```
preserve
keep if female==1
sum readingscore
restore
```

6. Temporary variables and temporary files – Stata can create variables and data files that exist only while your do-file is running. These are useful if you have no need to retain that variable or dataset. You must first create them using the `tempvar` or `tempfile` command; you can then use the temporarily variable or filename later by surrounding it in single quotes (like local macros). Some simple examples are below:

```
tempvar agetimes2
gen `agetimes2' = age*2
sum `agetimes2'
```

```

tempfile boys
tempfile girls
preserve
keep if female==1
save `girls'
restore
preserve
keep if female==0
save `boys'
restore

```

7. Scalars – a scalar is a name assigned to a number or a string (usually a number). When referencing a scalar, no quotes are required. For example:

```

scalar cpi = 1/1.3256
display cpi
gen realgdp = gdp*cpi

```

8. Saved results – Stata typically holds results in memory (as scalars) after certain commands are run. For example, after the `summarize` command, the mean is saved as `r(mean)`. Type `return list` after a command to see which results have been saved as scalars. The saved result is lost when you do something new; if you want to retain that number for later, create a scalar that contains it.

```

summarize income
return list
display r(mean)
display r(sd)
scalar meaninc=r(mean)    Saving the mean as a scalar for use later

```

9. Matrices – matrices are objects with `r` rows and `c` columns. The notation `a[r, c]` describes a matrix with `r` rows and `c` columns. E.g. `a[16, 16]`. Stata has a full matrix language called Mata. You can also easily create and manipulate matrices yourself (beyond the scope of this handout). Type `help matrix` for a list of commands.
10. Draw random samples from your data – if you would like to draw a random sample of size `n` from your existing dataset, use the commands `sample` or `bsample`. `sample` is a random sample *without replacement*, while `bsample` (bootstrap sample) is a random sample *with replacement*. Note that Stata keeps the sampled observations and drops all others, so you may want to use these with `preserve` and `restore`.

<code>sample 25, count</code>	draws a random sample of <code>n=25</code>
<code>sample 25</code>	draws a random sample of 25% of your cases
<code>bsample 10</code>	draws a random sample of <code>n=10</code> <i>with replacement</i> (bootstrap)

These commands have options for drawing more complex sampling methods (i.e., not a simple random sample) including stratified and cluster sampling.

11. Stata uses a pseudo-random number generator when selecting random samples and drawing random numbers from a distribution (see item 13 below). What this means is that the results are not truly random but generated by a mathematical algorithm. If you start with the same “seed” value, you will perfectly replicate the same random draws time and time again. This is valuable for replicating work that involves random numbers. To select a seed value include a `set seed` command at the beginning of your do-file. The seed value can be anything you choose:

```
set seed 12345
```

12. Add empty observations to a dataset – the command `set obs #` will change the number of observations in the current data. If the dataset is empty, it will create # blank observations that you can then fill in (say, if you are generating simulated data). If the dataset already has `_N` observations, then # must be at least as large as `_N`.

```
set obs 1000
```

13. Drawing random values from a distribution – Stata can draw random values from a long list of discrete and continuous probability distributions. Some common ones are below. (In each case the variable that will contain the random values is called `x`).

<code>gen x = rnormal()</code>	standard normal (0, 1)
<code>gen x = rnormal(<i>m</i>, <i>s</i>)</code>	normal (<i>m</i> , <i>s</i>)
<code>gen x = runiform()</code>	uniform distribution on (0, 1)
<code>gen x = runiform(<i>a</i>, <i>b</i>)</code>	uniform distribution on (<i>a</i> , <i>b</i>)
<code>gen x = rbinomial(<i>n</i>, <i>p</i>)</code>	binomial distribution (# of successes in <i>n</i> trials with <i>p</i> probability of success)
<code>gen x = rt(<i>df</i>)</code>	t-distribution with <i>df</i> degrees of freedom

In Stata, type `help random` for a complete list of random number functions.